__NOTOC__
原点の座標に直方体を作る。
centerがtrueの際、原点を中心に作られ、falseの際はx, y, z 共に正の方向に作られる。
引数が下記のパラメータと同じ順ならば、引数の名前を省略し、値だけを入力してもよい。
引数
size : 十進数または、長さ3の配列。ひとつの値が与えられた際、その値を辺長として持つ立方体となる。配列が与えられた際、それぞれ、x, y, z方向の長さとなる。初期値は1。
center : ブール型。オブジェクトの位置を決める。trueの際、原点を中心に作られる。falseの際はx, y, z 共に正の方向に作られる。初期値はfalse
例:
 
 cube(size = 1, center = false);
 cube(size = [1,2,3], center = true);
Openscad-cube.jpg
原点を中心に球を作る。引数名は省略可である。
引数
r :十進数。球の半径。解像度は大きさや、変数$fa, $fs, $fnに基づく。詳しくはOpenSCAD_User_Manual/Other_Language_Features
$fa : 最小分割角度（digree）
$fs : 最小分割長さ（mm）
$fn : 分割数
例:
 sphere(r = 1);
 sphere(r = 5);
 sphere(r = 10);
 //高解像度の半径2mm球
 sphere(2, $fn=100); 
 // これも高解像度の半径2mm球であるが、
 // 軸周りに多数の三角形を生成しない。
 sphere(2, $fa=5, $fs=0.1); 
Openscad-sphere.jpg
座標系の原点に円筒を生成する。
1つの半径 (r) を指定すると円筒になり、2つの異なる半径 (r1, r2) を指定すると円錐になる。
引数
h   数値。円筒の高さ。デフォルトは1。
r1: 数値。円筒底面の半径。デフォルトは1。
r2: 数値。円筒上面の半径。デフォルトは1.
r : 数値。円筒の上面と底面の半径。太さが均一な円筒を生成する場合はこの値を指定する。デフォルトは1。
center: boolean値。trueであれば円筒・円錐の高さの中央が原点となる。
デフォルトは false、つまり円筒の底面 (半径 r1) が原点となる。
$fa : 分割角度 (度)。
$fs : 分割辺の長さ (mm)。
例:
 cylinder(h = 10, r1 = 10, r2 = 20, center = false);
 cylinder(h = 10, r1 = 20, r2 = 10, center = true);
 cylinder(h = 10, r=20);
 cylinder(h = 10, r=20, $fs=6);
Openscad-cylinder.jpg
Create a polyhedron with a list of points and a list of triangles. The point list is all the vertexes of the shape, the triangle list is how the points relates to the surfaces of the polyhedron. 
Parameters
points : vector of points or vertexes (each a 3 vector).
triangles : vector of point triplets (each a 3 number vector). Each number is the 0-indexed point number from the point vector.
convexity: Integer. The convexity parameter specifies the maximum number of front sides (back sides) a ray intersecting the object might penetrate. This parameter is only needed for correctly displaying the object in OpenCSG preview mode and has no effect on the polyhedron rendering.
Syntax example
 polyhedron(points = [ [x, y, z], ... ], triangles = [ [p1, p2, p3..], ... ], convexity = N);
Triangle points ordering
When looking at the face from the outside inwards, the points must be clockwise. You can rearrange the order of the points or the order they are referenced in each triangle triple. The order of triangles is immaterial. Note that if your polygons are not all oriented the same way OpenSCAD will either print an error or crash completely, so pay attention to the vertex ordering.
Example, a square base pyramid:
polyhedron(
  points=[ [10,10,0],[10,-10,0],[-10,-10,0],[-10,10,0], // the four points at base
           [0,0,10]  ],                                 // the apex point 
  triangles=[ [0,1,4],[1,2,4],[2,3,4],[3,0,4],          // each triangle side
              [1,0,3],[2,1,3] ]                         // two triangles for square base
 );
frame|none|A simple polyhedron, square based pyramid
Ordering of triangle points
An example of a more complex polyhedron, and showing how to fix polyhedrons with badly oriented polygons.
When you select 'Thrown together' from the view menu and compile the design
(not compile and render!) you will see a preview with the mis-oriented polygons highlighted. Unfortunately this highlighting is not possible in the OpenCSG preview mode because it would interfere with the way the OpenCSG preview mode is implemented.)
Below you can see the code and the picture of such a problematic polyhedron, the bad polygons (triangles or compositions of triangles) are in pink.
// Bad polyhedron
polyhedron
    (points = [
	       [0, -10, 60], [0, 10, 60], [0, 10, 0], [0, -10, 0], [60, -10, 60], [60, 10, 60], 
	       [10, -10, 50], [10, 10, 50], [10, 10, 30], [10, -10, 30], [30, -10, 50], [30, 10, 50]
	       ], 
     triangles = [
		  [0,2,3],   [0,1,2],  [0,4,5],  [0,5,1],   [5,4,2],  [2,4,3],
                  [6,8,9],  [6,7,8],  [6,10,11], [6,11,7], [10,8,11],
		  [10,9,8], [0,3,9],  [9,0,6], [10,6, 0],  [0,4,10],
                  [3,9,10], [3,10,4], [1,7,11],  [1,11,5], [1,7,8],  
                  [1,8,2],  [2,8,11], [2,11,5]
		  ]
     );
frame|none|Polyhedron with badly oriented polygons
A correct polyhedron would be the following:
polyhedron
    (points = [
	       [0, -10, 60], [0, 10, 60], [0, 10, 0], [0, -10, 0], [60, -10, 60], [60, 10, 60], 
	       [10, -10, 50], [10, 10, 50], [10, 10, 30], [10, -10, 30], [30, -10, 50], [30, 10, 50]
	       ], 
     triangles = [
		  [0,3,2],  [0,2,1],  [4,0,5],  [5,0,1],  [5,2,4],  [4,2,3],
                  [6,8,9],  [6,7,8],  [6,10,11],[6,11,7], [10,8,11],
		  [10,9,8], [3,0,9],  [9,0,6],  [10,6, 0],[0,4,10],
                  [3,9,10], [3,10,4], [1,7,11], [1,11,5], [1,8,7],  
                  [2,8,1],  [8,2,11], [5,11,2]
		  ]
     );
Beginner's tip: 
If you don't really understand "orientation", try to identify the mis-oriented pink triangles and then permute the references to the points vectors until you get it right. E.g. in the above example, the third triangle (''[0,4,5]'') was wrong and we fixed it as ''[4,0,5]''. In addition, you may select "Show Edges" from the "View Menu", print a screen capture and number both the points and the triangles.  In our example, the points are annotated in black and the triangles in blue. Turn the object around and make a second copy from the back if needed. This way you can keep track.
Clockwise Technique: 
Orientation is determined by clockwise indexing. This means that if you're looking at the triangle (in this case [4,0,5]) from the outside you'll see that the path is clockwise around the center of the face. The winding order [4,0,5] is clockwise and therefore good.  The winding order [0,4,5] is counter-clockwise and therefore bad. Likewise, any other clockwise order of [4,0,5] works: [5,4,0] & [0,5,4] are good too. If you use the clockwise technique, you'll always have your faces outside (outside of OpenSCAD, other programs do use counter-clockwise as the outside though).
Think of it as a Left Hand Rule:
If you hold the triangle and the fingers of your hand curls is the same order as the points, then your thumb points outwards.
frame|none|Polyhedron with badly oriented polygons
Succinct description of a 'Polyhedron'
 * Points define all of the points/vertices in the shape.
 * Triangles is a list of triangles that connect up the points/vertices. 
Each point, in the point list, is defined with a 3-tuple x,y,z position specification. Points in the point list are automatically given an identifier starting at zero for use in the triangle list (0,1,2,3,... etc).
Each triangle, in the triangle list, is defined by selecting 3 of the points (using the point identifier) out of the point list.
e.g. triangles=[ [0,1,2] ] defines a triangle from the first point (points are zero referenced) to the second point and then to the third point.
When looking at any triangle from the outside, the triangle must list their 3 points in a clockwise order.